Pensieve header: Computing the Jones polynomial, further details and a faster procedure.

The Jones Polynomial of the Trefoil Knot




```
<< KnotTheory`
```

```
Loading KnotTheory` version of September 6, 2014, 13:37:37.2841.
Read more at http://katlas.org/wiki/KnotTheory.
```

```
K = Knot[3, 1]
```

```
Knot[3, 1]
```

```
PD[K]
```

KnotTheory: Loading precomputed data in PD4Knots`.

```
PD[X[1, 4, 2, 5], X[3, 6, 4, 1], X[5, 2, 6, 3]]
```

```
t1 = PD[K] /. x_X :→ If[PositiveQ[x], Xp @@ x, Xm @@ x]
```

```
PD[Xm[1, 4, 2, 5], Xm[3, 6, 4, 1], Xm[5, 2, 6, 3]]
```

```
t2 = Times @@ t1
```

```
Xm[1, 4, 2, 5] Xm[3, 6, 4, 1] Xm[5, 2, 6, 3]
```

```
t3 = t2 /. {
   Xp[i_, j_, k_, l_] :→ q * P[i, j] P[k, l] - q^2 * P[i, l] P[j, k],
   Xm[i_, j_, k_, l_] :→ -q^(-2) * P[i, j] P[k, l] + 1/q * P[i, l] P[j, k]
  }
```

$$\left(\frac{P[1,5]\,P[2,4]}{q}-\frac{P[1,4]\,P[2,5]}{q^2}\right)\left(\frac{P[2,6]\,P[3,5]}{q}-\frac{P[2,5]\,P[3,6]}{q^2}\right)\left(-\frac{P[1,4]\,P[3,6]}{q^2}+\frac{P[1,3]\,P[4,6]}{q}\right)$$

```
t4 = Expand[t3]
```

$$-\frac{P[1,4]\,P[1,5]\,P[2,4]\,P[2,6]\,P[3,5]\,P[3,6]}{q^4}+\frac{P[1,4]^2\,P[2,5]\,P[2,6]\,P[3,5]\,P[3,6]}{q^5}+\frac{P[1,4]\,P[1,5]\,P[2,4]\,P[2,5]\,P[3,6]^2}{q^5}-\frac{P[1,4]^2\,P[2,5]^2\,P[3,6]^2}{q^6}+\frac{P[1,3]\,P[1,5]\,P[2,4]\,P[2,6]\,P[3,5]\,P[4,6]}{q^3}-\frac{P[1,3]\,P[1,4]\,P[2,5]\,P[2,6]\,P[3,5]\,P[4,6]}{q^4}-\frac{P[1,3]\,P[1,5]\,P[2,4]\,P[2,5]\,P[3,6]\,P[4,6]}{q^4}+\frac{P[1,3]\,P[1,4]\,P[2,5]^2\,P[3,6]\,P[4,6]}{q^5}$$

```
SetAttributes[P, Orderless]
```

```
t5 = t4 //. P[a_, b_] P[b_, c_] :> P[a, c]
```

$$-\frac{P[1,4]^2\,P[2,5]^2\,P[3,6]^2}{q^6}+\frac{P[3,6]^2\,P[4,5]^2}{q^5}+\frac{P[2,5]^2\,P[4,6]^2}{q^5}+\frac{P[3,5]^2\,P[4,6]^2}{q^3}-\frac{3\,P[5,6]^2}{q^4}+\frac{P[1,4]^2\,P[5,6]^2}{q^5}$$

```
t6 = t5 /. {
    P[i_, i_] :> (q + 1/q),
    P[i_, j_]^2 :> (q + 1/q)
  }
```

$$-\frac{3\left(\frac{1}{q}+q\right)}{q^4}+\frac{3\left(\frac{1}{q}+q\right)^2}{q^5}+\frac{\left(\frac{1}{q}+q\right)^2}{q^3}-\frac{\left(\frac{1}{q}+q\right)^3}{q^6}$$

```
Expand[t6]
```

$$-\frac{1}{q^9}+\frac{1}{q^5}+\frac{1}{q^3}+\frac{1}{q}$$

```
Simplify[t6]
```

$$\frac{-1+q^4+q^6+q^8}{q^9}$$

```
Jones[Knot[4, 1]][q]
```

KnotTheory: Loading precomputed data in Jones4Knots`.

$$1+\frac{1}{q^2}-\frac{1}{q}-q+q^2$$

$$\text{Simplify}\left[\frac{t6}{q+q^{-1}}\ /.\ q\to\sqrt{q}\right]$$

$$\frac{-1+q+q^3}{q^4}$$

A Jones Polynomial Program

```
J[K_] := Module[{t1, t2, t3, t4, t5, t6, P},
  SetAttributes[P, Orderless];
  t1 = PD[K] /. x_X :> If[PositiveQ[x], Xp @@ x, Xm @@ x];
  t2 = Times @@ t1;
  t3 = t2 /. {
     Xp[i_, j_, k_, l_] :> q * P[i, j] P[k, l] - q^2 * P[i, l] P[j, k],
     Xm[i_, j_, k_, l_] :> -q^(-2) * P[i, j] P[k, l] + 1/q * P[i, l] P[j, k]
    };
  t4 = Expand[t3];
  t5 = t4 //. P[a_, b_] P[b_, c_] :> P[a, c];
  t6 = t5 /. {
     P[i_, i_] :> (q + 1/q),
     P[i_, j_]^2 :> (q + 1/q)
    };
  Simplify[t6/(q + q^-1) /. q -> Sqrt[q]]
 ]
```

```
J[Knot[8, 17]]
```

$$7 + \frac{1}{q^4} - \frac{3}{q^3} + \frac{5}{q^2} - \frac{6}{q} - 6\,q + 5\,q^2 - 3\,q^3 + q^4$$

```
Jones[Knot[8, 17]][q]
```

$$7 + \frac{1}{q^4} - \frac{3}{q^3} + \frac{5}{q^2} - \frac{6}{q} - 6\,q + 5\,q^2 - 3\,q^3 + q^4$$

```
Timing[
 (K ↦ Expand[J[K]] == Jones[K][q]) /@ AllKnots[{3, 10}]
]
```

```
{54.2188, {True, True, True, True, True, True, True, True, True, True, True, True,
  True, True, True, True, True, True, True, True, True, True, True, True, True, True,
  True, True, True, True, True, True, True, True, True, True, True, True, True, True,
  True, True, True, True, True, True, True, True, True, True, True, True, True, True,
  True, True, True, True, True, True, True, True, True, True, True, True, True, True,
  True, True, True, True, True, True, True, True, True, True, True, True, True, True,
  True, True, True, True, True, True, True, True, True, True, True, True, True, True,
  True, True, True, True, True, True, True, True, True, True, True, True, True, True,
  True, True, True, True, True, True, True, True, True, True, True, True, True, True,
  True, True, True, True, True, True, True, True, True, True, True, True, True, True,
  True, True, True, True, True, True, True, True, True, True, True, True, True, True,
  True, True, True, True, True, True, True, True, True, True, True, True, True, True,
  True, True, True, True, True, True, True, True, True, True, True, True, True, True,
  True, True, True, True, True, True, True, True, True, True, True, True, True, True,
  True, True, True, True, True, True, True, True, True, True, True, True, True, True,
  True, True, True, True, True, True, True, True, True, True, True, True, True, True,
  True, True, True, True, True, True, True, True, True, True, True, True, True, True,
  True, True, True, True, True, True, True, True, True, True, True, True, True}}
```

Testing the Reidemeister Moves

```
SetAttributes[P, Orderless];
TJ[t2_] := Module[{t3, t4, t5, t6},
  t3 = t2 /. {
     Xp[i_, j_, k_, l_] :> q * P[i, j] P[k, l] - q^2 * P[i, l] P[j, k],
     Xm[i_, j_, k_, l_] :> -q^(-2) * P[i, j] P[k, l] + 1/q * P[i, l] P[j, k]
    };
  t4 = Expand[t3];
  t5 = t4 //. P[a_, b_] P[b_, c_] :> P[a, c];
  t6 = t5 /. {
     P[i_, i_] :> (q + 1/q),
     P[i_, j_]^2 :> (q + 1/q)
    };
  Simplify[t6]
 ]
```



```
Xp[2, 2, 3, 1] // TJ
```

```
P[1, 3]
```

```
Xm[1, 2, 2, 3] // TJ
```

```
P[1, 3]
```




```
TJ /@ {Xp[1, 5, 2, 4] Xm[2, 5, 3, 6], Xp[4, 3, 5, 2] Xm[5, 1, 6, 2]}
```

```
{P[1, 3] P[4, 6], P[1, 3] P[4, 6]}
```




```
TJ /@ {Xp[4, 2, 5, 1] Xp[7, 3, 8, 2] Xp[8, 6, 9, 5], Xp[7, 5, 8, 4] Xp[8, 2, 9, 1] Xp[5, 3, 6, 2]}
```

$$\{q^3 \left(P[1,9] \left(P[3,7]\,P[4,6] - q\,P[3,6]\,P[4,7]\right) + q \left(q\,P[1,3]\,P[4,7]\,P[6,9] + P[1,4] \left(-P[3,7]\,P[6,9] + q\,P[3,6]\,P[7,9]\right)\right)\right),$$
$$q^3 \left(P[1,9] \left(P[3,7]\,P[4,6] - q\,P[3,6]\,P[4,7]\right) + q \left(q\,P[1,3]\,P[4,7]\,P[6,9] + P[1,4] \left(-P[3,7]\,P[6,9] + q\,P[3,6]\,P[7,9]\right)\right)\right)\}$$

```
%[[1]] == %[[2]]
```

```
True
```

Analyzing a knot suggested by David Vincent




```
K1 = PD[X[6, 14, 7, 13], X[14, 6, 15, 5], X[4, 7, 5, 8], X[8, 15, 9, 16],
   X[16, 3, 1, 4], X[12, 9, 13, 10], X[10, 1, 11, 2], X[2, 11, 3, 12]];
```

```
J1 = J[K1]
```

$$\frac{1-3 q+4 q^2-5 q^3+6 q^4-5 q^5+4 q^6-2 q^7+q^8}{q^7}$$

```
Select[AllKnots[{3, 8}], J[#] == J1 &]
```

{Knot[8, 14]}

A Faster Jones Program

```
FJ[K_] := Module[{todo, touched = {}, x, J = 1},
  SetAttributes[P, Orderless];
  todo = List @@ PD[K] /. x_X ⧴ If[PositiveQ[x], Xp @@ x, Xm @@ x];
  While[Length[todo] > 0,
   x = First@MaximalBy[todo, Length[touched ⋂ (List @@ #)] &];
   J *= x /. {
      Xp[i_, j_, k_, l_] ⧴ q * P[i, j] P[k, l] - q^2 * P[i, l] P[j, k],
      Xm[i_, j_, k_, l_] ⧴ -q^(-2) * P[i, j] P[k, l] + 1 / q * P[i, l] P[j, k]
     };
   J = Expand[J] //. P[a_, b_] P[b_, c_] ⧴ P[a, c] /. {
      P[i_, i_] ⧴ (q + 1 / q),
      P[i_, j_]^2 ⧴ (q + 1 / q)
     };
   todo = DeleteCases[todo, x];
  ];
  Simplify[J / (q + q^-1) /. q → √q]
 ]
```

```
FJ[Knot[3, 1]]
```

$$\frac{-1+q+q^3}{q^4}$$

Some Torus Knot Computations





```
Timing[
  (K ↦ Expand[FJ[K]] == Jones[K][q]) /@ AllKnots[{3, 10}]
]
```

```
{8.64063, {True, True, True, True, True, True, True, True, True, True, True, True,
  True, True, True, True, True, True, True, True, True, True, True, True, True, True,
  True, True, True, True, True, True, True, True, True, True, True, True, True, True,
  True, True, True, True, True, True, True, True, True, True, True, True, True, True,
  True, True, True, True, True, True, True, True, True, True, True, True, True, True,
  True, True, True, True, True, True, True, True, True, True, True, True, True, True,
  True, True, True, True, True, True, True, True, True, True, True, True, True, True,
  True, True, True, True, True, True, True, True, True, True, True, True, True, True,
  True, True, True, True, True, True, True, True, True, True, True, True, True, True,
  True, True, True, True, True, True, True, True, True, True, True, True, True, True,
  True, True, True, True, True, True, True, True, True, True, True, True, True, True,
  True, True, True, True, True, True, True, True, True, True, True, True, True, True,
  True, True, True, True, True, True, True, True, True, True, True, True, True, True,
  True, True, True, True, True, True, True, True, True, True, True, True, True, True,
  True, True, True, True, True, True, True, True, True, True, True, True, True, True,
  True, True, True, True, True, True, True, True, True, True, True, True, True, True,
  True, True, True, True, True, True, True, True, True, True, True, True, True, True,
  True, True, True, True, True, True, True, True, True, True, True, True, True}}
```

```
? TorusKnots
```

TorusKnots[n_] returns a list of all torus knots with up to n crossings.

```
Do[
 {t1, J1} = Timing@J[K];
 {t2, J2} = Timing@FJ[K];
 Print[{K, Length@PD[K], J1 == J2, t1, t2}],
 {K, TorusKnots[100]}
]
```

```
{TorusKnot[3, 2], 3, True, 0., 0.}
{TorusKnot[5, 2], 5, True, 0., 0.015625}
{TorusKnot[7, 2], 7, True, 0., 0.015625}
{TorusKnot[4, 3], 8, True, 0.046875, 0.015625}
{TorusKnot[9, 2], 9, True, 0.171875, 0.015625}
{TorusKnot[5, 3], 10, True, 0.28125, 0.015625}
{TorusKnot[11, 2], 11, True, 1.375, 0.}
{TorusKnot[13, 2], 13, True, 6.51563, 0.015625}
{TorusKnot[7, 3], 14, True, 10.5781, 0.03125}
{TorusKnot[5, 4], 15, True, 24.5156, 0.0625}
{TorusKnot[15, 2], 15, True, 39.5313, 0.015625}
{TorusKnot[8, 3], 16, True, 71.7188, 0.03125}
{TorusKnot[17, 2], 17, True, 371.859, 0.078125}
$Aborted
```

```
Do[
 {t2, J2} = Timing@FJ[K];
 Print[{K, Length@PD[K], t2}],
 {K, TorusKnots[100]}
]
```

```
{TorusKnot[3, 2], 3, 0.}
{TorusKnot[5, 2], 5, 0.015625}
{TorusKnot[7, 2], 7, 0.03125}
{TorusKnot[4, 3], 8, 0.046875}
{TorusKnot[9, 2], 9, 0.015625}
{TorusKnot[5, 3], 10, 0.046875}
{TorusKnot[11, 2], 11, 0.015625}
{TorusKnot[13, 2], 13, 0.0625}
{TorusKnot[7, 3], 14, 0.109375}
{TorusKnot[5, 4], 15, 0.203125}
```

```
{TorusKnot[15, 2], 15, 0.046875}
{TorusKnot[8, 3], 16, 0.109375}
{TorusKnot[17, 2], 17, 0.09375}
{TorusKnot[19, 2], 19, 0.109375}
{TorusKnot[10, 3], 20, 0.21875}
{TorusKnot[7, 4], 21, 0.375}
{TorusKnot[21, 2], 21, 0.09375}
{TorusKnot[11, 3], 22, 0.21875}
{TorusKnot[23, 2], 23, 0.109375}
{TorusKnot[6, 5], 24, 1.28125}
{TorusKnot[25, 2], 25, 0.171875}
{TorusKnot[13, 3], 26, 0.21875}
{TorusKnot[9, 4], 27, 0.671875}
{TorusKnot[27, 2], 27, 0.234375}
{TorusKnot[7, 5], 28, 1.89063}
{TorusKnot[14, 3], 28, 0.28125}
{TorusKnot[29, 2], 29, 0.203125}
{TorusKnot[31, 2], 31, 0.28125}
{TorusKnot[8, 5], 32, 2.6875}
{TorusKnot[16, 3], 32, 0.3125}
{TorusKnot[11, 4], 33, 1.17188}
{TorusKnot[33, 2], 33, 0.15625}
{TorusKnot[17, 3], 34, 0.34375}
{TorusKnot[7, 6], 35, 9.}
{TorusKnot[35, 2], 35, 0.1875}
{TorusKnot[9, 5], 36, 3.3125}
{TorusKnot[37, 2], 37, 0.28125}
{TorusKnot[19, 3], 38, 0.46875}
{TorusKnot[13, 4], 39, 1.23438}
{TorusKnot[39, 2], 39, 0.359375}
{TorusKnot[20, 3], 40, 0.578125}
{TorusKnot[41, 2], 41, 0.34375}
{TorusKnot[43, 2], 43, 0.359375}
{TorusKnot[11, 5], 44, 5.45313}
{TorusKnot[22, 3], 44, 0.75}
{TorusKnot[15, 4], 45, 1.82813}
```

```
{TorusKnot[45, 2], 45, 0.421875}
{TorusKnot[23, 3], 46, 0.671875}
{TorusKnot[47, 2], 47, 0.609375}
{TorusKnot[8, 7], 48, 89.3438}
{TorusKnot[12, 5], 48, 8.5}
{TorusKnot[49, 2], 49, 0.5625}
{TorusKnot[25, 3], 50, 1.1875}
{TorusKnot[17, 4], 51, 3.0625}
{TorusKnot[51, 2], 51, 0.6875}
{TorusKnot[13, 5], 52, 9.96875}
{TorusKnot[26, 3], 52, 0.953125}
{TorusKnot[53, 2], 53, 0.578125}
{TorusKnot[9, 7], 54, 132.422}
{TorusKnot[11, 6], 55, 40.4688}
{TorusKnot[55, 2], 55, 0.953125}
{TorusKnot[14, 5], 56, 13.1875}
{TorusKnot[28, 3], 56, 1.4375}
{TorusKnot[19, 4], 57, 3.79688}
{TorusKnot[57, 2], 57, 0.921875}
{TorusKnot[29, 3], 58, 1.65625}
{TorusKnot[59, 2], 59, 0.828125}
{TorusKnot[10, 7], 60, 140.563}
{TorusKnot[61, 2], 61, 0.515625}
{TorusKnot[31, 3], 62, 0.890625}
{TorusKnot[9, 8], 63, 650.813}
{TorusKnot[21, 4], 63, 4.21875}
{TorusKnot[63, 2], 63, 0.984375}
{TorusKnot[16, 5], 64, 16.4844}
{TorusKnot[32, 3], 64, 1.625}
{TorusKnot[13, 6], 65, 55.8125}
{TorusKnot[65, 2], 65, 0.4375}
{TorusKnot[11, 7], 66, 91.8594}
{TorusKnot[67, 2], 67, 0.40625}
{TorusKnot[17, 5], 68, 6.92188}
{TorusKnot[34, 3], 68, 0.625}
{TorusKnot[23, 4], 69, 2.03125}
```

```
{TorusKnot[69, 2], 69, 0.40625}
{TorusKnot[35, 3], 70, 0.65625}
{TorusKnot[71, 2], 71, 0.421875}
{TorusKnot[12, 7], 72, 104.641}
{TorusKnot[18, 5], 72, 7.84375}
{TorusKnot[73, 2], 73, 0.4375}
{TorusKnot[37, 3], 74, 0.8125}
{TorusKnot[25, 4], 75, 2.23438}
{TorusKnot[75, 2], 75, 0.484375}
{TorusKnot[19, 5], 76, 8.98438}
{TorusKnot[38, 3], 76, 0.8125}
{TorusKnot[11, 8], 77, 447.031}
{TorusKnot[77, 2], 77, 0.421875}
{TorusKnot[13, 7], 78, 151.781}
{TorusKnot[79, 2], 79, 0.578125}
{TorusKnot[10, 9], 80, 1607.05}
{TorusKnot[40, 3], 80, 0.65625}
{TorusKnot[27, 4], 81, 1.90625}
{TorusKnot[81, 2], 81, 0.390625}
{TorusKnot[41, 3], 82, 0.65625}
{TorusKnot[83, 2], 83, 0.40625}
{TorusKnot[21, 5], 84, 8.625}
{TorusKnot[17, 6], 85, 35.3906}
{TorusKnot[85, 2], 85, 0.515625}
{TorusKnot[43, 3], 86, 0.921875}
{TorusKnot[29, 4], 87, 2.78125}
{TorusKnot[87, 2], 87, 0.546875}
{TorusKnot[11, 9], 88, 1710.84}
{TorusKnot[22, 5], 88, 9.5625}
{TorusKnot[44, 3], 88, 0.78125}
{TorusKnot[89, 2], 89, 0.484375}
{TorusKnot[15, 7], 90, 141.031}
{TorusKnot[13, 8], 91, 573.219}
{TorusKnot[91, 2], 91, 0.484375}
{TorusKnot[23, 5], 92, 9.625}
{TorusKnot[46, 3], 92, 0.859375}
```

```
{TorusKnot[31, 4], 93, 2.5}
{TorusKnot[93, 2], 93, 0.515625}
{TorusKnot[47, 3], 94, 0.875}
{TorusKnot[19, 6], 95, 42.3281}
{TorusKnot[95, 2], 95, 0.53125}
{TorusKnot[16, 7], 96, 165.703}
{TorusKnot[24, 5], 96, 11.7813}
{TorusKnot[97, 2], 97, 0.5625}
{TorusKnot[49, 3], 98, 1.04688}
{TorusKnot[11, 10], 99, 8210.92}
{TorusKnot[33, 4], 99, 2.95313}
{TorusKnot[99, 2], 99, 0.59375}
{TorusKnot[50, 3], 100, 1.}
```